INTÉGRATION : SOLUTIONS DES EXERCICES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M.1

Enoncé

Calculer les intégrales définies suivantes :

1. $\int_1^2 (x-1)(x-2)\,\mathrm{d}x$
2. $\int_0^1 \left(3\sqrt{x}-4x\right)\,\mathrm{d}x$
3. $\int_0^8 \left(\sqrt{2x}+\sqrt[3]{x}\right)\,\mathrm{d}x$
4. $\int_{-2}^{-1} x^{-3}\,\mathrm{d}x$
5. $\int_{-1}^{+1} \left(x^2-\sqrt{|x|}\right)\sqrt[3]{x^2}\,\mathrm{d}x$
6. $\int_a^b \left(\frac{1}{\sqrt{x}}-x^2\right)\,\mathrm{d}x$
7. $\int_2^0 \sqrt{|1-x|}\,\mathrm{d}x$
8. $\int_0^1 \frac{1}{1+x}\,\mathrm{d}x$
9. $\int_{-\infty}^{+\infty} e^t\,\mathrm{d}t$
10. $\int_2^3 a^x\,\mathrm{d}x$ pour $a \in \mathbb{R}_+^* - \{1\}$
11. $\int_{-1}^{+1} e^{-2x+1}\,\mathrm{d}x$

Solution

Pour toutes les questions, il s'agit de calculer une intégrale définie de forme générale

$\int_a^b f(x)\,dx$ où a et b sont éventuellement infinis. On utilise `int` sous la forme `int(f(x),x=a..b)` où `f(x)` est une expression en `x`. Le taux de réussite de Maple est de 100%.

```
> id1:=int((x-1)*(x-2),x=1..2);
```

$$id1 := -\frac{1}{6} \tag{1.1}$$

```
> id2:=int(3*x^(1/2)-4*x,x=0..3);
```

$$id2 := 6\sqrt{3} - 18 \tag{1.2}$$

```
> id3:=int(sqrt(2*x)+x^(1/3),x=0..8);
```

$$id3 := \frac{100}{3} \tag{1.3}$$

```
> id4:=int(x^(-3),x=-2..-1);
```

$$id4 := -\frac{3}{8} \tag{1.4}$$

```
> id5:=int(x^2-abs(x)^(2/3),x=-1..1);
```

$$id5 := -\frac{8}{15} \tag{1.5}$$

```
> id6:=int(x^(-1/2)-x^2,x=a..b);
```

$$id6 := -2\sqrt{a} + \frac{1}{3}a^3 + 2\sqrt{b} - \frac{1}{3}b^3 \tag{1.6}$$

```
> id7:=int(abs(1-x)^(1/2),x=2..0);
```

$$id7 := -\frac{4}{3} \tag{1.7}$$

```
> id8:=int((1+x)^(-1),x=0..1);
```

$$id8 := \ln(2) \tag{1.8}$$

```
> id9:=int(exp(x),x=-infinity..infinity);
```

$$id9 := \infty \tag{1.9}$$

```
> id10:=int(a^x,x=2..3);
```

$$id10 := \frac{a^2(-1+a)}{\ln(a)} \tag{1.10}$$

```
> id11:=int(exp(-2*x+1),x=-1..1);
```

$$id11 := \frac{1}{2}e^3 - \frac{1}{2}e^{-1} \tag{1.11}$$

Exercice M.2

Enoncé

L'objectif de cet exercice est de calculer l'intégrale définie $\int_{-2}^{2} \left(x^3 - \sqrt[3]{x}\right) x^2 \, dx$.

1. Calculer directement l'intégrale avec l'instruction `int`.

2. Critiquez le résultat renvoyé par Maple.
3. Comment obtenir le résultat juste avec Maple?

Solution

1. L'utilisation directe de `int` donne :

```
> int((x^3-x^(1/3))*x^2,x=-2..2);
```

$$-\frac{18}{5}\,2^{1/3}-\frac{6}{5}\,2^{1/3}\sqrt{-3} \tag{2.1}$$

soit encore, en forçant l'évaluation :

```
> evalf(%);
```

$$-4.535715780-2.618696728\,\mathrm{I} \tag{2.2}$$

2. L'intégrale définie d'une fonction réelle continue ne peut pas être un nombre complexe. Le résultat renvoyé par Maple est donc faux. Si on remarque que la fonction $f(x)=\left(x^3-x^{\frac{1}{3}}\right)x^2$ est impaire (vérifiez que $f(-x)=-f(x)$), alors elle est symétrique par rapport à 0 et le résultat exact est $\int_{-2}^{2}\left(x^3-\sqrt[3]{x}\right)x^2\,\mathrm{d}x=0$.

3. Il faut faire comprendre à Maple que $\sqrt[3]{x}$ est un réel - n'est pas complexe - quand x est négatif. Plusieurs possibilités existent pour contourner la difficulté. On en évoque deux ici. La première consiste à redéfinir "à la main" la fonction de départ en distinguant la demie droite numérique positive et la demie droite numérique négative. On crée alors une fonction continue par morceaux.

```
> xp:=piecewise(x>=0,(x^3-x^(1/3))*x^2,(-abs(x)^3+abs(x)^(1/3))*
  x^2);#redéfinition de la fonction intégrande
  int(xp,x=-2..2);#calcul de l'intégrale définie
```

$$xp:=\begin{cases}\left(x^3-x^{1/3}\right)x^2 & 0\le x\\ \left(-|x|^3+|x|^{1/3}\right)x^2 & otherwise\end{cases}$$

$$0 \tag{2.3}$$

Cette fois, le résultat est juste. On pouvait également l'obtenir en chargeant le paquetage `RealDomain`, qui oblige Maple à "travailler" dans le corps des réels.

```
> with(RealDomain);
  int((x^3-x^(1/3))*x^2,x=-2..2);#calcul direct de l'intégrale
  définie
```

[$\Im$, $\Re$, \`^\`, arccos, arccosh, arccot, arccoth, arccsc, arccsch, arcsec, arcsech, arcsin, arcsinh, arctan, arctanh, cos, cosh, cot, coth, csc, csch, *eval*, exp, *expand*, *limit*, ln, log, sec, sech, signum, *simplify*, sin, sinh, *solve*, sqrt, surd, tan, tanh]

$$0 \tag{2.4}$$

Exercice M.3

Enoncé

Calculer les intégrales définies suivantes à l'aide d'un changement de variable :

1. $\int_{1}^{5}\sqrt{2x-1}\,\mathrm{d}x$

2. $\displaystyle\int_0^3 (25-3x)^{-\frac{1}{2}}\,dx$

3. $\displaystyle\int_0^2 \frac{x^2}{2+x^3}\,dx$

4. $\displaystyle\int_1^2 \frac{1}{x(1+\ln(x))}\,dx$

5. $\displaystyle\int_2^3 \frac{x+\frac{1}{2}}{x^2+x-3}\,dx$

6. $\displaystyle\int_0^1 \frac{e^x}{(10-3e^x)^2}\,dx$

7. $\displaystyle\int_0^4 \frac{1}{1+\sqrt{x}}\,dx$

8. $\displaystyle\int_0^1 \frac{e^x-e^{-x}}{e^x+e^{-x}}\,dx$

9. $\displaystyle\int_0^1 x^2\sqrt{1+x^3}\,dx$

10. $\displaystyle\int_{\ln(3)}^{\ln(8)} \sqrt{e^x+1}\,dx$

Solution

L'énoncé suggère de répondre aux questions à l'aide de la commande **changevar** du paquetage **student**, ou, alternativement, avec la commande **Change** du paquetage **IntegrationTools**. Dans les deux cas, on commence par poser l'intégrale à calculer sous sa forme inerte. Puis on effectue le changement de variable, qui renvoie une forme inerte. Il ne reste plus qu'à demander l'évaluation avec l'instruction **evalf**. Soulignons que Maple laisse à l'appréciation de l'utilisateur ce que doit être le changement de variable. Autrement dit, il faut du flair.

Remarque : la commande **Change** n'accepte pas toutes les transformations proposées sous forme d'équations simples. Le message suivant est alors affiché :

```
Error, (in IntegrationTools:-Change) expected lhs of transformation equations of type And(name, Not(constant)), or (unknown) function. Received `+`
```

On contourne la difficulté en encadrant le membre de gauche de l'équation par des guillemets (simple quote), ce qui a pour effet de lui donner le type name. L'évaluation est correcte mais le changement de variable n'est pas affiché. Morale : la commande **changevar** reste plus intéressante que la commande **Change**.

```
> restart;
```

```
with(student):with(IntegrationTools):#chargement des paquetages
```

1.

```
> id1:=Int((2*x-1)^(1/2),x=1..5);#écriture de l'intégrale sous forme inerte
changevar((2*x-1)^(1/2)=u,id1,u);#changement de variable dans le paquetage student
evalf(%);#évaluation
Change(id1,(2*x-1)^(1/2)=u);#changement de variable dans le paquetage IntegrationTools
evalf(%);#évaluation
```

$$id1 := \int_{1}^{5} \sqrt{2\,x-1}\,\mathrm{d}x$$

$$\int_{1}^{\sqrt{9}} u^2\,\mathrm{d}u$$

$$8.666666667$$

$$\int_{1}^{3} u^2\,\mathrm{d}u$$

$$8.666666667 \qquad (3.1)$$

Pour les questions suivantes, on utilisera toujours la même grille méthodologique : écriture de l'intégrale; résolution avec **changevar**; résolution avec **Change**.

2.

```
> id2:=Int((25-3*x)^(-1/2),x=0..3);
changevar((25-3*x)^(-1/2)=u,id2,u);
evalf(%);
Change(id2,(25-3*x)^(-1/2)=u);
evalf(%);
```

$$id2 := \int_{0}^{3} \frac{1}{\sqrt{25-3\,x}}\,\mathrm{d}x$$

$$\int_{\frac{1}{25}\sqrt{25}}^{\frac{1}{16}\sqrt{16}} \left(\frac{50}{3} - \frac{2}{3}\,\frac{25\,u^2-1}{u^2}\right)\mathrm{d}u$$

$$0.6666666667$$

$$\frac{2}{3}\int_{\frac{1}{5}}^{\frac{1}{4}} \frac{1}{u^2}\,\mathrm{d}u$$

(3.2)

$$0.6666666667 \tag{3.2}$$

3.

```
> id3:=Int(x^2/(2+x^3),x=0..2);
  id3a:=changevar(2+x^3=u,id3,u);
  evalf(id3a);
  id3b:=Change(id3,2+x^3=u);#le membre de gauche n'est pas du type name
  id3c:=Change(id3,`2+x^3`=u);#correction du membre de gauche
  evalf(id3c);
```

$$id3 := \int_0^2 \frac{x^2}{2+x^3}\,dx$$

$$id3a := \int_2^{10} \frac{1}{3\,u}\,du$$

$$0.5364793041$$

Error, (in IntegrationTools:-Change) expected lhs of transformation equations of type And(name, Not(constant)), or (unknown) function. Received `+`

$$id3c := \int_0^2 \frac{x^2}{2+x^3}\,dx$$

$$0.5364793041 \tag{3.3}$$

4.

```
> id4:=Int(1/(x*(1+log(x))),x=1..2);
  changevar(1+log(x)=u,id4,u);
  evalf(%);
  Change(id4,1+log(x)=u);
  evalf(%);
```

$$id4 := \int_1^2 \frac{1}{x\,(1+\ln(x))}\,dx$$

$$\int_1^{1+\ln(2)} \frac{1}{u}\,du$$

$$0.5265890341$$

$$\int_1^{1+\ln(2)} \frac{1}{u}\,du$$

$$0.5265890341 \tag{3.4}$$

5.

```
> id5:=Int((x+0.5)/(x^2+x-3),x=2..3);
```

```
id5a:=changevar(x^2+x-3=u,id5,u);
evalf(id5a);
id5b:=Change(id5,`x^2+x-3`=u);
evalf(id5b);
```

$$id5 := \int_{2}^{3} \frac{x+0.5}{x^2+x-3}\, dx$$

$$id5a := \int_{3}^{9} \frac{1}{2\,u}\, du$$

$$0.5493061443$$

$$id5b := \int_{2}^{3} \frac{x+0.5000000000}{x^2+x-3.}\, dx$$

$$0.5493061443 \quad \textbf{(3.5)}$$

6.

```
> id6:=Int(exp(x)/(10-3*exp(x))^2,x=0..1);
changevar(10-3*exp(x)=u,id6,u);
evalf(%);
Change(id6,10-3*exp(x)=u);
evalf(%);
```

$$id6 := \int_{0}^{1} \frac{e^x}{\left(10-3\,e^x\right)^2}\, dx$$

$$\int_{10-3\,e}^{7} \frac{1}{3\,u^2}\, du$$

$$0.1330342964$$

$$\frac{1}{3}\int_{10-3\,e}^{7} \frac{1}{u^2}\, du$$

$$0.1330342964 \quad \textbf{(3.6)}$$

7.

```
> id7:=Int(1/(1+sqrt(x)),x=0..4);
changevar(1+sqrt(x)=u,id7,u);
evalf(%);
Change(id7,1+sqrt(x)=u);
evalf(%);
```

$$id7 := \int_{0}^{4} \frac{1}{1+\sqrt{x}}\, dx$$

$$\int_{1}^{1+\sqrt{4}} \frac{2\sqrt{u^2-2u+1}}{u}\,\mathrm{d}u$$

$$1.802775423$$

$$2\left(\int_{1}^{3} \frac{u-1}{u}\,\mathrm{d}u\right)$$

$$1.802775423 \tag{3.7}$$

8.

```
> id8:=Int((exp(x)-exp(-x))/(exp(x)+exp(-x)),x=0..1);
  id8a:=changevar(exp(x)+exp(-x)=u,id8,u);
  evalf(id8a);
  id8b:=Change(id8,`exp(x)+exp(-x)`=u);
  evalf(id8b);
```

$$id8 := \int_{0}^{1} \frac{e^{x}-e^{-x}}{e^{x}+e^{-x}}\,\mathrm{d}x$$

$$id8a := \int_{2}^{e+e^{-1}} \frac{1}{u}\,\mathrm{d}u$$

$$0.4337808305$$

$$id8b := \int_{0}^{1} \frac{e^{x}-e^{-x}}{e^{x}+e^{-x}}\,\mathrm{d}x$$

$$0.4337808305 \tag{3.8}$$

9.

```
> id9:=Int(x^2*sqrt(1+x^3),x=0..1);
  id9a:=changevar(1+x^3=u,id9,u);
  evalf(id9a);
  id9b:=Change(id9,`1+x^3`=u);
  evalf(id9b);
```

$$id9 := \int_{0}^{1} x^2\sqrt{1+x^3}\,\mathrm{d}x$$

$$id9a := \int_{1}^{2} \frac{1}{3}\sqrt{u}\,\mathrm{d}u$$

$$0.4063171388$$

$$id9b := \int_{0}^{1} x^2\sqrt{1+x^3}\,\mathrm{d}x$$

$$0.4063171388 \tag{3.9}$$

10.

```
> id10:=Int(sqrt(exp(x)+1),x=log(3)..log(8));
  changevar(exp(x)+1=u^2,id10,u);
  evalf(%);
  Change(id10,exp(x)+1=u^2);
  evalf(%);
```

$$id10 := \int_{\ln(3)}^{3\ln(2)} \sqrt{e^x + 1}\, dx$$

$$\int_{\sqrt{4}}^{\sqrt{9}} \frac{2u^2}{-1+u^2}\, du$$

$$2.405465108$$

$$\int_{2}^{3} \frac{2\sqrt{u^2}\, u}{-1+u^2}\, du$$

$$2.405465108 \qquad \textbf{(3.10)}$$

Exercice M.4

Enoncé

Calculer par parties les intégrales définies suivantes :

1. $\displaystyle\int_2^3 \ln(x^2-1)\, dx$
2. $\displaystyle\int_0^1 \frac{x+1}{e^x}\, dx$
3. $\displaystyle\int_0^1 x\, 2^x\, dx$
4. $\displaystyle\int_1^2 \ln\left(\frac{x}{x+1}\right) dx$
5. $\displaystyle\int_{-1}^{+1} (x+e^x)(2x-e^{-x})\, dx$
6. $\displaystyle\int_2^3 (x+1)\, a^{x+1}\, dx$ avec $a \in \mathbb{R}_+^* - \{1\}$
7. $\displaystyle\int_2^1 x\, [\ln(x)]^2\, dx$
8. $\displaystyle\int_1^4 \frac{\ln(x)}{\sqrt{x}}\, dx$

Solution

L'énoncé suggère d'effectuer des intégrations par parties à l'aide de la commande **intparts** du paquetage **student** ou, alternativement, avec la commande **Parts** du paquetage **IntegrationTools**. Dans les deux cas, on commence par poser l'intégrale à calculer sous sa forme inerte. Puis on effectue le changement de variable, qui renvoie une forme inerte. Il ne reste plus qu'à demander l'évaluation avec l'instruction **evalf**.

```
> restart;
  with(student):with(IntegrationTools):#chargement des paquetages
```

1.

```
> id1:=Int(log(x^2-1),x=2..3);
  intparts(id1,log(x^2-1));evalf(%);
  Parts(id1,log(x^2-1));evalf(%);
```

$$id1 := \int_2^3 \ln\left(x^2-1\right)\, dx$$

$$9\ln(2) - 2\ln(3) - \left(\int_2^3 \frac{2x^2}{x^2-1}\, dx\right)$$

$$1.635634939$$

$$9\ln(2) - 2\ln(3) - \left(\int_2^3 \frac{2x^2}{x^2-1}\, dx\right)$$

$$1.635634939 \tag{4.1}$$

2.

```
> id2:=Int((x+1)/exp(x),x=0..1);
  intparts(id2,x+1);evalf(%);
  Parts(id2,x+1);evalf(%);
```

$$id2 := \int_0^1 \frac{x+1}{e^x}\, dx$$

$$-2\,e^{-1} + 1 - \left(\int_0^1 \left(-e^{-x}\right) dx\right)$$

$$0.8963616764$$

$$-2\,e^{-1} + 1 - \left(\int_0^1 \left(-\frac{1}{e^x}\right) dx\right)$$

$$0.8963616764 \tag{4.2}$$

3.

```
> id3:=Int(x*2^x,x=0..1);
  intparts(id3,x);evalf(%);
  Parts(id3,x);evalf(%);
```

$$id3 := \int_0^1 x\, 2^x \, dx$$

$$\frac{2}{\ln(2)} - \left(\int_0^1 \frac{2^x}{\ln(2)} \, dx\right)$$

$$0.804021101$$

$$\frac{2}{\ln(2)} - \left(\int_0^1 \frac{2^x}{\ln(2)} \, dx\right)$$

$$0.804021101 \quad \textbf{(4.3)}$$

4.

```
> id4:=Int(log(x/(x+1)),x=2..3);
  intparts(id4,log(x/(x+1)));evalf(%);
  Parts(id4,log(x/(x+1)));evalf(%);
```

$$id4 := \int_2^3 \ln\left(\frac{x}{x+1}\right) dx$$

$$3\ln\left(\frac{3}{4}\right) - 2\ln\left(\frac{2}{3}\right) - \left(\int_2^3 \left(\frac{1}{x+1} - \frac{x}{(x+1)^2}\right)(x+1)\, dx\right)$$

$$-0.3397980738$$

$$5\ln(3) - 8\ln(2) - \left(\int_2^3 \left(\frac{1}{x+1} - \frac{x}{(x+1)^2}\right)(x+1)\, dx\right)$$

$$-0.3397980725 \quad \textbf{(4.4)}$$

5.

```
> id5:=Int((x+exp(x))*(2*x-exp(-x)),x=-1..1);
  intparts(id5,x+exp(x));evalf(%);
  Parts(id5,x+exp(x));evalf(%);
```

$$id5 := \int_{-1}^1 (x+e^x)(2x-e^{-x})\, dx$$

$$(1+e)(1+e^{-1}) - (-1+e^{-1})(1+e) - \left(\int_{-1}^1 (1+e^x)(x^2+e^{-x})\, dx\right)$$

$$1.540609978$$

$$(1+e)(1+e^{-1}) - (-1+e^{-1})(1+e) - \left(\int_{-1}^1 (1+e^x)(x^2+e^{-x})\, dx\right)$$

$$1.540609978 \quad \textbf{(4.5)}$$

6.

```
> id6:=Int((x+1)*a^(x+1),x=2..3);
  intparts(id6,x+1);
```

```
Parts(id6,x+1);
```

$$id6 := \int_2^3 (x+1)\, a^{x+1}\, dx$$

$$\frac{4\,a^4}{\ln(a)} - \frac{3\,a^3}{\ln(a)} - \left(\int_2^3 \frac{a^{x+1}}{\ln(a)}\, dx\right)$$

$$\frac{4\,a^4}{\ln(a)} - \frac{3\,a^3}{\ln(a)} - \left(\int_2^3 \frac{a^{x+1}}{\ln(a)}\, dx\right) \tag{4.6}$$

7.

```
> id7:=Int(x*(log(x))^2,x=2..3);
intparts(id7,log(x)^2);evalf(%);
Parts(id7,log(x)^2);evalf(%);
```

$$id7 := \int_2^3 x \ln(x)^2\, dx$$

$$\frac{9}{2} \ln(3)^2 - 2 \ln(2)^2 - \left(\int_2^3 \ln(x)\, x\, dx\right)$$

$$2.162903363$$

$$\frac{9}{2} \ln(3)^2 - 2 \ln(2)^2 - \left(\int_2^3 \ln(x)\, x\, dx\right)$$

$$2.162903363 \tag{4.7}$$

8.

```
> id1:=Int(log(x)/sqrt(x),x=2..3);
intparts(id1,log(x));evalf(%);
Parts(id1,log(x));evalf(%);
```

$$id1 := \int_2^3 \frac{\ln(x)}{\sqrt{x}}\, dx$$

$$2 \ln(3) \sqrt{3} - 2 \ln(2) \sqrt{2} - \left(\int_2^3 \frac{2}{\sqrt{x}}\, dx\right)$$

$$0.573839338$$

$$2 \ln(3) \sqrt{3} - 2 \ln(2) \sqrt{2} - \left(\int_2^3 \frac{2}{\sqrt{x}}\, dx\right)$$

$$0.573839338 \tag{4.8}$$

Exercice M.5

Enoncé

Donner les primitives des fonctions suivantes :

1. $f(x) = \frac{1+x^5-x^6}{1-x}$
2. $f(x) = \frac{3x^2+1}{x^2(x^2+1)^2}$
3. $f(x) = \frac{x^{3+1}}{x^2-x+2}$
4. $f(x) = \frac{1}{x^2-5x+4}$
5. $f(x) = \frac{x}{x^3-3x+2}$
6. $f(x) = \frac{1}{x^3-7x+6}$
7. $f(x) = \frac{7x-5}{x^3+x^2-6x}$
8. $f(x) = \frac{ax+b}{(x-c)^2}$
9. $f(x) = \frac{x^2+6x+5}{x^2-6x+5}$
10. $f(x) = \frac{2x+1}{(x^2-1)^2}$
11. $f(x) = \frac{4x^2-6x+1}{2x^3-x^2}$
12. $f(x) = \frac{2x^4+1}{x^3(x^2+x+1)}$

Solution

Cet exercice porte sur l'intégration de fonctions rationnelles comme on peut s'en assurer avec `infolevel[integrate]:=2`.

1.

```
> restart;
  X1:=(1+x^5-x^6)/(1-x);
  infolevel[integrate]:=2;
  Int(X1,x)=int(X1,x);
```

$$X1 := \frac{1+x^5-x^6}{1-x}$$

$$infolevel_{int} := 2$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{1+x^5-x^6}{1-x}\,dx = \frac{1}{6}x^6 - \ln(x-1) \quad \textbf{(5.1)}$$

2.

```
> X2:=(3*x^2+1)/(x^2*(x^2+1)^2);
  Int(X2,x)=int(X2,x);
```

$$X2 := \frac{3\,x^2+1}{x^2\,(x^2+1)^2}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{3\,x^2+1}{x^2\,(x^2+1)^2}\,dx = -\frac{1}{x} + \frac{x}{x^2+1} \tag{5.2}$$

3.

```
> X3:=x^(3+1)/(x^2-x+2);
  Int(X3,x)=int(X3,x);
```

$$X3 := \frac{x^4}{x^2-x+2}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{x^4}{x^2-x+2}\,dx = \frac{1}{3}\,x^3 + \frac{1}{2}\,x^2 - x - \frac{3}{2}\ln(x^2-x+2) + \frac{1}{7}\sqrt{7}\arctan\left(\frac{1}{7}\,(2\,x-1)\sqrt{7}\right) \tag{5.3}$$

4.

```
> X4:=1/(x^2-5*x+4);
  Int(X4,x)=int(X4,x);
```

$$X4 := \frac{1}{x^2-5\,x+4}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{1}{x^2-5\,x+4}\,dx = \frac{1}{3}\ln(x-4) - \frac{1}{3}\ln(x-1) \tag{5.4}$$

5.

```
> X5:=x/(x^3-3*x+2);
  Int(X5,x)=int(X5,x);
```

$$X5 := \frac{x}{x^3-3\,x+2}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{x}{x^3-3\,x+2}\,dx = -\frac{2}{9}\ln(x+2) + \frac{2}{9}\ln(x-1) - \frac{1}{3\,(x-1)} \tag{5.5}$$

6.

```
> X6:=1/(x^3-7*x+6);
```

```
Int(X6,x)=int(X6,x);
```

$$X6 := \frac{1}{x^3 - 7x + 6}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{1}{x^3 - 7x + 6}\, dx = -\frac{1}{4}\ln(x-1) + \frac{1}{5}\ln(-2+x) + \frac{1}{20}\ln(x+3) \tag{5.6}$$

7.

```
> X7:=(7*x-5)/(x^3+x^2-6*x);
  Int(X7,x)=int(X7,x);
```

$$X7 := \frac{7x - 5}{x^3 + x^2 - 6x}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{7x-5}{x^3 + x^2 - 6x}\, dx = \frac{9}{10}\ln(-2+x) - \frac{26}{15}\ln(x+3) + \frac{5}{6}\ln(x) \tag{5.7}$$

8.

```
> X8:=(ax+b)/(x-c)^2;
  Int(X8,x)=int(X8,x);
```

$$X8 := \frac{ax + b}{(x - c)^2}$$

```
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
```

$$\int \frac{ax+b}{(x-c)^2}\, dx = -\frac{ax+b}{x-c} \tag{5.8}$$

9.

```
> X9:=(x^2+6*x+5)/(x^2-6*x+5);
  Int(X9,x)=int(X9,x);
```

$$X9 := \frac{x^2 + 6x + 5}{x^2 - 6x + 5}$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{x^2 + 6x + 5}{x^2 - 6x + 5}\, dx = x + 15\ln(-5+x) - 3\ln(x-1) \tag{5.9}$$

10.

```
> X10:=(2*x+1)/(x^2-1)^2;
  Int(X10,x)=int(X10,x);
```

$$X10 := \frac{2x + 1}{(x^2 - 1)^2}$$

```
int/indef1: first-stage indefinite integration
```

int/ratpoly: rational function integration

$$\int \frac{2x+1}{(x^2-1)^2}\,dx = \frac{1}{4}\ln(x+1) - \frac{1}{4}\ln(x-1) - \frac{3}{4(x-1)} + \frac{1}{4(x+1)} \tag{5.10}$$

11.

```
> X11:=(4*x^2-6*x+1)/(2*x^3-x^2);
  Int(X11,x)=int(X11,x);
```

$$X11 := \frac{4x^2-6x+1}{2x^3-x^2}$$

int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration

$$\int \frac{4x^2-6x+1}{2x^3-x^2}\,dx = -2\ln(2x-1) + \frac{1}{x} + 4\ln(x) \tag{5.11}$$

12.

```
> X12:=(2*x^4+1)/(x^3*(x^2+x+1));
  Int(X12,x)=int(X12,x);
```

$$X12 := \frac{2x^4+1}{x^3(x^2+x+1)}$$

int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration

$$\int \frac{2x^4+1}{x^3(x^2+x+1)}\,dx = \frac{1}{x} + \ln(x^2+x+1) - \frac{1}{2x^2} \tag{5.12}$$

Exercice M.6

<u>Enoncé</u>

Donner les primitives des fonctions suivantes :

1. $f(x) = \dfrac{x-1}{\sqrt{2x+1}}$
2. $f(x) = \dfrac{x^2}{\sqrt{x-1}} + 2x\sqrt{x-1}$
3. $f(x) = \dfrac{x^2\sqrt{x}}{1+\sqrt{x}}$
4. $f(x) = \dfrac{x}{2+\sqrt{1+x}}$
5. $f(x) = \dfrac{x}{\sqrt[3]{1+x} - \sqrt{1+x}}$
6. $f(x) = \dfrac{x+\sqrt{x+2}+\sqrt{x+1}}{\sqrt{x+2}+\sqrt{x+1}}$
7. $f(x) = \dfrac{1}{\sqrt{x^2+x+1}}$

8. $f(x) = \dfrac{1}{(1+x^2)^{\frac{3}{2}}}$

9. $f(x) = \dfrac{1}{x^2\sqrt{x^2-x+1}}$

10. $f(x) = \dfrac{1}{(x-1)\sqrt{-x^2+3x-2}}$

11. $f(x) = \dfrac{\sqrt{1+x^2}+1}{\sqrt{1+x^2}-1}$

12. $f(x) = \dfrac{x}{(1+x^2)\sqrt{1-x^4}}$

Solution

Les intégrandes contiennent des radicaux.

```
> restart;
  infolevel[integrate]:=2:#demande d'informations sur les étapes
  de la résolution
```

1.

```
> X1:=(x-1)/sqrt(2*x+1);
  Int(X1,x)=int(X1,x);
```

$$X1 := \frac{x-1}{\sqrt{2x+1}}$$

$$\int \frac{x-1}{\sqrt{2x+1}}\,\mathrm{d}x = \frac{1}{3}\sqrt{2x+1}\,(-4+x) \tag{6.1}$$

2.

```
> X2:=x^2/sqrt(x-1)+2*x*sqrt(x-1);
  Int(X2,x)=int(X2,x);
```

$$X2 := \frac{x^2}{\sqrt{x-1}} + 2x\sqrt{x-1}$$

$$\int \left(\frac{x^2}{\sqrt{x-1}} + 2x\sqrt{x-1}\right)\mathrm{d}x = \frac{2}{15}\sqrt{x-1}\,(4+2x+9x^2) \tag{6.2}$$

3.

```
> X3:=x^2*sqrt(x)/(1+sqrt(x));
  Int(X3,x)=int(X3,x);
```

$$X3 := \frac{x^{5/2}}{1+\sqrt{x}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
```

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{x^{5/2}}{1+\sqrt{x}}\,dx = \frac{1}{3}x^3 - \frac{2}{5}x^{5/2} + \frac{1}{2}x^2 - \frac{2}{3}x^{3/2} + x - 2\sqrt{x} + 2\ln\left(1+\sqrt{x}\right) \tag{6.3}$$

4.

```
> X4:=x/(2+sqrt(1+x));
  Int(X4,x)=int(X4,x);
```

$$X4 := \frac{x}{2+\sqrt{1+x}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{x}{2+\sqrt{1+x}}\,dx = \frac{2}{3}(1+x)^{3/2} - 2x - 2 + 6\sqrt{1+x} - 12\ln\left(2+\sqrt{1+x}\right) \tag{6.4}$$

5.

```
> X5:=x/((1+x)^(1/3)-sqrt(1+x));
  Int(X5,x)=int(X5,x);
```

$$X5 := \frac{x}{(1+x)^{1/3} - \sqrt{1+x}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
```

$$\int \frac{x}{(1+x)^{1/3} - \sqrt{1+x}}\,dx = -\frac{2}{3}(1+x)^{3/2} - \frac{3}{4}(1+x)^{4/3} - \frac{6}{7}(1+x)^{7/6} - x - 1 - \frac{6}{5}(1+x)^{5/6} - \frac{3}{2}(1+x)^{2/3} \tag{6.5}$$

6.

```
> X6:=(x+sqrt(x+2)+sqrt(1+x))/(sqrt(x+2)+sqrt(1+x));
  Int(X6,x)=int(X6,x);
```

$$X6 := \frac{x+\sqrt{x+2}+\sqrt{1+x}}{\sqrt{x+2}+\sqrt{1+x}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
```

```
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
```

$$\int \frac{x+\sqrt{x+2}+\sqrt{1+x}}{\sqrt{x+2}+\sqrt{1+x}}\,\mathrm{d}x = -\frac{2}{5}\,(1+x)^{5/2}+\frac{2}{3}\,(1+x)^{3/2}+\frac{2}{5}\,(x+2)^{5/2}-\frac{4}{3}\,(x+2)^{3/2}+x \tag{6.6}$$

7.

```
> X7:=1/sqrt(x^2+x+1);
  Int(X7,x)=int(X7,x);
```

$$X7 := \frac{1}{\sqrt{x^2+x+1}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
```

$$\int \frac{1}{\sqrt{x^2+x+1}}\,\mathrm{d}x = \operatorname{arcsinh}\left(\frac{2}{3}\sqrt{3}\left(x+\frac{1}{2}\right)\right) \tag{6.7}$$

8.

```
> X8:=1/(1+x^2)^(3/2);
  Int(X8,x)=int(X8,x);
```

$$X8 := \frac{1}{(1+x^2)^{3/2}}$$

$$\int \frac{1}{(1+x^2)^{3/2}}\,\mathrm{d}x = \frac{x}{\sqrt{1+x^2}} \tag{6.8}$$

9.

```
> X9:=1/(x^2*sqrt(x^2-x+1));
  Int(X9,x)=int(X9,x);
```

$$X9 := \frac{1}{x^2\sqrt{x^2-x+1}}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
```

$$\int \frac{1}{x^2\sqrt{x^2-x+1}}\,\mathrm{d}x = -\frac{\sqrt{x^2-x+1}}{x}-\frac{1}{2}\operatorname{arctanh}\left(\frac{1}{2}\,\frac{2-x}{\sqrt{x^2-x+1}}\right) \tag{6.9}$$

10.

```
> X10:=1/((x-1)*sqrt(-x^2+3*x-2));
  Int(X10,x)=int(X10,x);
```

$$X10 := \frac{1}{(-1+x)\sqrt{-x^2+3x-2}}$$

$$\int \frac{1}{(-1+x)\sqrt{-x^2+3x-2}}\,dx = \frac{2(x-2)}{\sqrt{-x^2+3x-2}} \qquad (6.10)$$

11.

```
> X11:=(sqrt(1+x^2)+1)/(sqrt(1+x^2)-1);
  Int(X11,x)=int(X11,x);
```

$$X11 := \frac{\sqrt{1+x^2}+1}{\sqrt{1+x^2}-1}$$

```
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
```

$$\int \frac{\sqrt{1+x^2}+1}{\sqrt{1+x^2}-1}\,dx = x - \frac{2}{x} - \frac{2\left(1+x^2\right)^{3/2}}{x} + 2x\sqrt{1+x^2} + 2\,\mathrm{arcsinh}(x) \qquad (6.11)$$

12.

```
> X12:=x/((1+x^2)*sqrt(-x^4+1));
  Int(X12,x)=int(X12,x);
```

$$X12 := \frac{x}{\left(1+x^2\right)\sqrt{-x^4+1}}$$

$$\int \frac{x}{\left(1+x^2\right)\sqrt{-x^4+1}}\,dx = \frac{1}{2}\,\frac{(-1+x)(1+x)}{\sqrt{-x^4+1}} \qquad (6.12)$$

Exercice M.7

Enoncé

Donner les primitives des fonctions suivantes :

1. $f(x) = \dfrac{\sqrt{x}+\ln(x)}{x}$

2. $f(x) = \dfrac{1}{\sqrt{\left(1+x^2\right)\ln\left(x+\sqrt{1+x^2}\right)}}$

3. $f(x) = \dfrac{[\ln(x)]^2}{x^2}$

4. $f(x) = \frac{(a^x - b^x)^2}{a^x b^x}$
5. $f(x) = \frac{1}{2^x + 3}$
6. $f(x) = \frac{e^{2x}}{\sqrt{e^x + 1}}$
7. $f(x) = \frac{1}{e^{2x} + e^x - 2}$
8. $f(x) = \left(e^{\frac{x}{a}} + e^{-\frac{x}{a}}\right)^2$
9. $f(x) = e^x \sqrt{a - be^x}$
10. $f(x) = e^{\sqrt{x}}$
11. $f(x) = \sqrt{1 - e^{-2x}}$
12. $f(x) = \left[\ln\left(x + \sqrt{1 + x^2}\right)\right]^2$
13. $f(x) = x^2 \ln\left(\sqrt{1 - x}\right)$
14. $f(x) = |x|$

Solution

On est face à un pot pourri d'intégrales difficiles, voire très difficiles. Il est prudent de vérifier la solution proposée par Maple en dérivant le résultat affiché et le comparer avec l'intégrande.

```
> restart;
  infolevel[integrate]:=2;#demande d'informations sur les étapes
  de la résolution
```

$$infolevel_{int} := 2 \qquad (7.1)$$

1.

```
> X1:=(sqrt(x)+ln(x))/x;
  Int(X1,x)=int(X1,x);#affichage du résultat
  diff(rhs(%),x);radnormal(%);#vérification du résultat
```

$$X1 := \frac{\sqrt{x} + \ln(x)}{x}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying algebraic substitution
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/ln: case of integrand containing ln
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
```

```
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
```

$$\int \frac{\sqrt{x}+\ln(x)}{x}\,dx = 2\sqrt{x}+\frac{1}{2}\ln(x)^2$$

$$\frac{1}{\sqrt{x}}+\frac{\ln(x)}{x}$$

$$\frac{\sqrt{x}+\ln(x)}{x} \tag{7.2}$$

2.

```
> X2:=1/sqrt((1+x^2)*ln(x+sqrt(1+x^2)));
  Int(X2,x)=int(X2,x);
```

$$X2 := \frac{1}{\sqrt{(1+x^2)\ln\left(x+\sqrt{1+x^2}\right)}}$$

$$\int \frac{1}{\sqrt{(1+x^2)\ln\left(x+\sqrt{1+x^2}\right)}}\,dx = \int \frac{1}{\sqrt{(1+x^2)\ln\left(x+\sqrt{1+x^2}\right)}}\,dx \tag{7.3}$$

En renvoyant la question posée, Maple signale qu'il est impuissant. Il faut l'aider pour le guider vers la solution. Ici, un changement de variable s'impose.

```
> with(student):#chargement du paquetage student
  I2a:=changevar(x+sqrt(1+x^2)=u,Int(X2,x),u);#changement de
  variable
  I2a=value(I2a);#calcul de la primitive
```

$$I2a := \int \frac{1}{\sqrt{\ln(u)}\,u}\,du$$

$$\int \frac{1}{\sqrt{\ln(u)}\,u}\,du = 2\sqrt{\ln(u)} \tag{7.4}$$

Il ne reste plus qu'à expliciter la primitive comme fonction de la variable *x* et vérifier le résultat.

```
> sol:=subs(u=x+sqrt(1+x^2),rhs(%));#primitive de X2
  diff(sol,x);radnormal(%);#vérification du résultat
```

$$sol := 2\sqrt{\ln\left(x+\sqrt{1+x^2}\right)}$$

$$\frac{1+\frac{x}{\sqrt{1+x^2}}}{\sqrt{\ln\left(x+\sqrt{1+x^2}\right)}\left(x+\sqrt{1+x^2}\right)}$$

$$\frac{1}{\sqrt{1+x^2}\sqrt{\ln\left(x+\sqrt{1+x^2}\right)}} \tag{7.5}$$

3.

```
> X3:=(ln(x))^2/x^2;
  Int(X3,x)=int(X3,x);
```

```
diff(rhs(%),x);
```

$$X3 := \frac{\ln(x)^2}{x^2}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/ln: case of integrand containing ln
```

$$\int \frac{\ln(x)^2}{x^2}\,dx = -\frac{\ln(x)^2}{x} - \frac{2\ln(x)}{x} - \frac{2}{x}$$

$$\frac{\ln(x)^2}{x^2} \tag{7.6}$$

4.

```
> X4:=(a^x-b^x)^2/(a^x*b^x);
  Int(X4,x)=int(X4,x);
  diff(rhs(%),x);radnormal(%);simplify(%,exp);
```

$$X4 := \frac{(a^x - b^x)^2}{a^x b^x}$$

$$\int \frac{(a^x - b^x)^2}{a^x b^x}\,dx = \frac{\frac{(e^{x\ln(a)})^2}{\ln(a)-\ln(b)} - \frac{(e^{x\ln(b)})^2}{\ln(a)-\ln(b)} - 2\,x\,e^{x\ln(a)}\,e^{x\ln(b)}}{e^{x\ln(a)}\,e^{x\ln(b)}}$$

$$\frac{1}{e^{x\ln(a)}\,e^{x\ln(b)}}\left(\frac{2\,(e^{x\ln(a)})^2\ln(a)}{\ln(a)-\ln(b)} - \frac{2\,(e^{x\ln(b)})^2\ln(b)}{\ln(a)-\ln(b)} - 2\,e^{x\ln(a)}\,e^{x\ln(b)} - 2\,x\ln(a)\,e^{x\ln(a)}\,e^{x\ln(b)} - 2\,x\,e^{x\ln(a)}\ln(b)\,e^{x\ln(b)}\right)$$

$$- \frac{\left(\frac{(e^{x\ln(a)})^2}{\ln(a)-\ln(b)} - \frac{(e^{x\ln(b)})^2}{\ln(a)-\ln(b)} - 2\,x\,e^{x\ln(a)}\,e^{x\ln(b)}\right)\ln(a)}{e^{x\ln(a)}\,e^{x\ln(b)}}$$

$$- \frac{\left(\frac{(e^{x\ln(a)})^2}{\ln(a)-\ln(b)} - \frac{(e^{x\ln(b)})^2}{\ln(a)-\ln(b)} - 2\,x\,e^{x\ln(a)}\,e^{x\ln(b)}\right)\ln(b)}{e^{x\ln(a)}\,e^{x\ln(b)}}$$

$$\frac{(e^{x\ln(a)})^2 - 2\,e^{x\ln(a)}\,e^{x\ln(b)} + (e^{x\ln(b)})^2}{e^{x\ln(a)}\,e^{x\ln(b)}}$$

$$\frac{(a^x)^2 - 2\,a^x b^x + (b^x)^2}{a^x b^x} \tag{7.7}$$

5.

```
> X5:=1/(2^x+3);
  Int(X5,x)=int(X5,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X5 := \frac{1}{2^x + 3}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \frac{1}{2^x + 3}\,\mathrm{d}x = -\frac{1}{3}\,\frac{\ln(2^x + 3)}{\ln(2)} + \frac{1}{3}\,\frac{\ln(2^x)}{\ln(2)}$$

$$-\frac{1}{3}\,\frac{2^x}{2^x + 3} + \frac{1}{3}$$

$$\frac{1}{2^x + 3} \tag{7.8}$$

6.

```
> X6:=exp(2*x)/sqrt(exp(x)+1);
  Int(X6,x)=int(X6,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X6 := \frac{e^{2x}}{\sqrt{e^x + 1}}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/exp: case of integrand containing exp
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
```

$$\int \frac{e^{2x}}{\sqrt{e^x + 1}}\,\mathrm{d}x = \frac{2}{3}\left(e^x + 1\right)^{3/2} - 2\sqrt{e^x + 1}$$

$$\sqrt{e^x + 1}\; e^x - \frac{e^x}{\sqrt{e^x + 1}}$$

$$\frac{e^{2x}}{\sqrt{e^x + 1}} \tag{7.9}$$

7.

```
> X7:=1/(exp(2*x)+exp(x)-2);
  Int(X7,x)=int(X7,x);
```

```
diff(rhs(%),x);normal(%,expanded);
```

$$X7 := \frac{1}{e^{2x}+e^{x}-2}$$

$$\int \frac{1}{e^{2x}+e^{x}-2}\,dx = \frac{1}{3}\ln\left(e^{x}-1\right)-\frac{1}{2}\ln\left(e^{x}\right)+\frac{1}{6}\ln\left(e^{x}+2\right)$$

$$\frac{1}{3}\frac{e^{x}}{e^{x}-1}-\frac{1}{2}+\frac{1}{6}\frac{e^{x}}{e^{x}+2}$$

$$\frac{1}{\left(e^{x}\right)^{2}+e^{x}-2} \tag{7.10}$$

8.

```
> X8:=(exp(x/a)+exp(-x/a))^2;
  Int(X8,x)=int(X8,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X8 := \left(e^{\frac{x}{a}}+e^{-\frac{x}{a}}\right)^{2}$$

$$\int \left(e^{\frac{x}{a}}+e^{-\frac{x}{a}}\right)^{2} dx = \frac{1}{2}\left(e^{\frac{x}{a}}\right)^{2} a+2x-\frac{1}{2}\frac{a}{\left(e^{\frac{x}{a}}\right)^{2}}$$

$$\left(e^{\frac{x}{a}}\right)^{2}+2+\frac{1}{\left(e^{\frac{x}{a}}\right)^{2}}$$

$$e^{\frac{2x}{a}}+2+e^{-\frac{2x}{a}} \tag{7.11}$$

9.

```
> X9:=exp(x)*sqrt(a-b*exp(x));
  Int(X9,x)=int(X9,x);
  diff(rhs(%),x);
```

$$X9 := e^{x}\sqrt{a-b\,e^{x}}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
```

$$\int e^{x}\sqrt{a-b\,e^{x}}\,dx = -\frac{2}{3}\frac{\left(a-b\,e^{x}\right)^{3/2}}{b}$$

$$e^{x}\sqrt{a-b\,e^{x}} \tag{7.12}$$

10.

```
> X10:=exp(sqrt(x));
```

```
Int(X10,x)=int(X10,x);
diff(rhs(%),x);
```

$$X10 := e^{\sqrt{x}}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/exp: case of integrand containing exp
```

$$\int e^{\sqrt{x}}\,dx = 2\,e^{\sqrt{x}}\sqrt{x} - 2\,e^{\sqrt{x}}$$

$$e^{\sqrt{x}} \tag{7.13}$$

11.

```
> X11:=sqrt(1-exp(-2*x));
  Int(X11,x)=int(X11,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X11 := \sqrt{1-e^{-2x}}$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying derivative-divides
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/algebraic2/algebraic: algebraic integration
int/algebraic2/algebraic: applying algebraic substitution
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\int \sqrt{1-e^{-2x}}\,dx = -\sqrt{1-e^{-2x}} - \frac{1}{2}\ln\left(\sqrt{1-e^{-2x}}-1\right) + \frac{1}{2}\ln\left(\sqrt{1-e^{-2x}}+1\right)$$

$$-\frac{e^{-2x}}{\sqrt{1-e^{-2x}}} - \frac{1}{2}\,\frac{e^{-2x}}{\sqrt{1-e^{-2x}}\left(\sqrt{1-e^{-2x}}-1\right)} + \frac{1}{2}\,\frac{e^{-2x}}{\sqrt{1-e^{-2x}}\left(\sqrt{1-e^{-2x}}+1\right)}$$

$$\sqrt{1-e^{-2x}} \tag{7.14}$$

12.

```
> X12:=(ln(x+sqrt(1+x^2)))^2;
  Int(X12,x)=int(X12,x);
```

```
diff(rhs(%),x);
```

$$X12 := \ln\left(x+\sqrt{1+x^2}\right)^2$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/ln: case of integrand containing ln
int/rischnorm: enter Risch-Norman integrator
int/rischnorm: exit Risch-Norman integrator
int/risch: enter Risch integration
```

$$\int \ln\left(x+\sqrt{1+x^2}\right)^2 \mathrm{d}x = \int \ln\left(x+\sqrt{1+x^2}\right)^2 \mathrm{d}x$$

$$\ln\left(x+\sqrt{1+x^2}\right)^2 \qquad \textbf{(7.15)}$$

13.

```
> X13:=x^2*ln(sqrt(1-x));
  Int(X13,x)=int(X13,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X13 := \frac{1}{2}\, x^2 \ln(1-x)$$

```
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef2: applying change of variables
int/indef1: first-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/ln: case of integrand containing ln
int/ln: case of integrand containing ln
int/ln: case of integrand containing ln
int/ln: case of integrand containing ln
```

$$\int \frac{1}{2}\, x^2 \ln(1-x)\, \mathrm{d}x = -\frac{1}{6}\,(1-x)^3 \ln(1-x) - \frac{1}{12}\,x^2 - \frac{1}{6}\,x + \frac{11}{36} - \frac{1}{18}\,x^3 + \frac{1}{2}\ln(1-x)\,(1-x)^2 - \frac{1}{2}\,(1-x)\ln(1-x)$$

$$\frac{1}{2}\ln(1-x)\,(1-x)^2 + \frac{1}{6}\,(1-x)^2 + \frac{1}{3}\,x - \frac{1}{6} - \frac{1}{6}\,x^2 - (1-x)\ln(1-x) + \frac{1}{2}\ln(1-x)$$

$$\frac{1}{2}\,x^2 \ln(1-x) \qquad \textbf{(7.16)}$$

14.

```
> X14:=abs(x);
  Int(X14,x)=int(X14,x);
  diff(rhs(%),x);simplify(%,symbolic);
```

$$X14 := |x|$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
```

$$\int |x| \, dx = \begin{cases} -\frac{1}{2} x^2 & x \leq 0 \\ \frac{1}{2} x^2 & 0 < x \end{cases}$$

$$\begin{cases} -x & x \leq 0 \\ x & 0 < x \end{cases}$$

$$\begin{cases} -x & x \leq 0 \\ x & 0 < x \end{cases} \quad (7.17)$$

Exercice M.8

Enoncé

Soit l'intégrale définie :

$I_n = \int_{-1}^{+1} (x^2 - 1)^n \, dx$ où n est un entier naturel.

1. Calculer I_0, I_1, I_2, I_3.
2. Trouver une relation de récurrence entre I_n et I_{n-1}.
3. Mettre alors au point une procédure pour calculer I_n (n quelconque).

Solution

Si on demande directement à Maple d'intégrer $I_n = \int_{-1}^{+1} (x^2 - 1)^n \, dx$, le résultat se réfère à la fonction Gamma.

```
> restart;
  int((x^2-1)^n,x=-1..1);
```

$$\frac{(-1)^n \Gamma(n+1) \sqrt{\pi}}{\Gamma\left(\frac{3}{2} + n\right)} \quad (8.1)$$

L'exercice veut montrer qu'il existe une méthode commode pour calculer l'intégrale parce qu'elle est en fait le $(n+1) - ième$ terme d'une suite relativement simple à mettre en évidence.

1. Après chargement des paquetages dont on aura besoin dans cette session, on définit la fonction `Itg` qui associe à tout entier naturel n l'intégrale défnie $I_n = \int_{-1}^{+1} (x^2 - 1)^n \, dx$. Le calcul des 4 premiers termes de la suite est immédiat.

```
> with(student):with(IntegrationTools):with(RealDomain):
  Itg:=n->Int((x^2-1)^n,x=-1..1);
```

```
Itg(0)=value(Itg(0));
Itg(1)=value(Itg(1));
Itg(2)=value(Itg(2));
Itg(3)=value(Itg(3));
```

$$Itg := n \to \int_{-1}^{1} (x^2-1)^n \, dx$$

$$\int_{-1}^{1} 1 \, dx = 2$$

$$\int_{-1}^{1} (x^2-1) \, dx = -\frac{4}{3}$$

$$\int_{-1}^{1} (x^2-1)^2 \, dx = \frac{16}{15}$$

$$\int_{-1}^{1} (x^2-1)^3 \, dx = -\frac{32}{35} \qquad \textbf{(8.2)}$$

2.Commençons par intégrer par parties $I_n = \int_{-1}^{+1} (x^2-1)^n \, dx$.

```
> Itgn:=simplify(intparts(Itg(n),(x^2-1)^n),symbolic);
```

$$Itgn := -2\,n \left(\int_{-1}^{1} x^2 (x^2-1)^{n-1} \, dx \right) \qquad \textbf{(8.3)}$$

Comme $(x^2-1)^{n-1} x^2 = (x^2-1)^{n-1} (x^2 - 1 + 1) = (x^2-1)^n + (x^2-1)^{n-1}$, on aura en définitive $I_n = -\frac{2\,n}{2\,n+1} I_{n-1}$

```
> Integn:=GetIntegrand(op(3,Itgn));#assignation de l'intégrande dans Itgn
is(Integn=(x^2-1+1)*(x^2-1)^(n-1));#test sur l'intégrande dans Itgn
Integn:=(x^2-1)^n+(x^2-1)^(n-1);#réécriture de l'intégrande dans Itgn
Itgn:=-2*n*Expand(Int(Integn,x=-1..1));#réécriture de Itgn
is(Itgn=-2*n*(Itg(n)+Itg(n-1)));#test sur Itgn
rec:=solve(Itg(n)=-2*n*(Itg(n)+Itg(n-1)),Itg(n));#
is(rec=-2*n/(2*n+1)*Itg(n-1));#test sur la relation de récurrence
```

$$Integn := x^2 (x^2-1)^{n-1}$$

$$true$$

$$Integn := (x^2-1)^n + (x^2-1)^{n-1}$$

$$Itgn := -2\,n \left(\int_{-1}^{1} (x^2-1)^n \,dx + \int_{-1}^{1} \frac{(x^2-1)^n}{x^2-1} \,dx \right)$$

$$true$$

$$rec := -\frac{2\,n \left(\int_{-1}^{1} (x^2-1)^{n-1} \,dx \right)}{1+2\,n}$$

$$true \qquad \textbf{(8.4)}$$

3. La procédure **integ** a pour argument unique l'entier **n** Elle est récursive.

```
> integ:=proc(n::integer)
  if n=0 then value(Itg(0));
  else -2*n/(2*n+1)*value(integ(n-1));#la procédure fait appel ..
  . à elle même.
  end if;
  end proc;
```

integ := **proc**(*n*::*integer*) **(8.5)**

if $n = 0$ **then** *value*(*Itg*(0)) **else** $-2 * n * value(integ(n-1)) / (2 * n + 1)$ **end if**

end proc

```
> seq(integ(i),i=0..10);#test de la procédure sur les 11 premiers
  termes de la suite.
```

$$2, -\frac{4}{3}, \frac{16}{15}, -\frac{32}{35}, \frac{256}{315}, -\frac{512}{693}, \frac{2048}{3003}, -\frac{4096}{6435}, \frac{65536}{109395}, -\frac{131072}{230945}, \frac{524288}{969969} \qquad \textbf{(8.6)}$$